

```
//local variables
VAR a:ARRAY[3..12][0..9] OF INTEGER
    i, j: INTEGER
```

In row- or column-major order,  
 base of array a has offset -400  
 offset of i is: -404  
 offset of j is: -408

The layout is illustrated below.



To load  $a[i][j]$  in row-major order, its address is

$$\&a[3][0] + ((i-3) * (9-0+1) + (j-0)) * \text{sizeofElement} \\ = (\$fp-400) + ((i-3) * 10 + j) * 4$$


```
add $s0, $fp, -400 // base of a
lw $s1, -404($fp) // load i
sub $s1, $s1, 3    // i-3
mul $s1, $s1, 10   // (i-3)*10
lw $s2, -408($fp) //load j
add $s1, $s1, $s2  // (i-3)*10+j
mul $s1, $s1, 4    // *4
add $s0, $s0, $s1  // add base
lw $s1, ($s0)      // load a[i][j]
```

To load  $a[i][j]$  in column-major order, its address is

$\&a[3][0] + ((j-0)*(12-3+1) + (i-3)) * \text{sizeofElement}$   
 $= (\$fp-400) + (j*10 + (i-3))*4$

```
add $s0, $fp, -400 // base of a
lw $s1, -408($fp) // load j
mul $s1, $s1, 10 // j*10
lw $s2, -404($fp) //load i
sub $s2, $s2, 3 // i-3
add $s1, $s1, $s2 // j*10+(i-3)
mul $s1, $s1, 4 // *4
add $s0, $s0, $s1 // add base
lw $s1, ($s0) // load a[i][j]
```

In row-pointer layout (see the picture below),  
 base of a has offset -40  
 i: -44  
 j: -48



The address of  $a[i][j]$  is  
 $a[i] + (j-0) * \text{sizeofElement}$   
 $= *(base\ a + (i-3) * 4) + j * 4$   
 $= *((\$fp - 40) + (i-3) * 4) + j * 4$

```
add $s0, $fp, -40    // base of a
lw $s1, -44($fp)    // load i
sub $s1, $s1, 3      // i-3
mul $s1, $s1, 4      // (i-3)*4
add $s0, $s0, $s1    //base a+ (i-3)*4 -> &a[i]
lw $s1, ($s0)        // load a[i]
lw $s0, -48($fp)     //load j
mul $s0, $s0, 4      // *4
add $s1, $s1, $s0    // add base (*a[i])
lw $s0, ($s1)        // load a[i][j]
```